We shall study an underdetermined system of PDEs coming from mathematical physics.

J. Wheeler: **Exist potentials for the Einstein equations?**

Pommaret (1995): **No.** See J.-F. Pommaret, *Dualité différentielles et applications*, C.R. Acad. Sci. Paris, Serie I 320 (1995). (See also H. Weyl, *Space Time Matter*, fourth edition, Dover, 1952, for more details on Einstein equations.)

Let us prove this by using *OreModules*!

```
> with(Ore_algebra):
> with(OreModules):
```

We define the Weyl algebra $Alg = A_4$ (where $x[1]$, $x[2]$, $x[3]$ stand for the three space components and $x[4] = c\,t$ for the time):

```
> Alg := DefineOreAlgebra(diff=[D[1],x[1]], diff=[D[2],x[2]],
> diff=[D[3],x[3]], diff=[D[4],x[4]], polynom={x[1],x[2],x[3],x[4]}):
```

The linearized Ricci equations in the vacuum are defined by the following matrix of differential operator.

```
> R := evalm([[D[2]^2+D[3]^2-D[4]^2, D[1]^2, D[1]^2, -D[1]^2, -2*D[1]*D[2], 0, 0,
> -2*D[1]*D[3], 0, 2*D[1]*D[4]],
> [D[2]^2, D[1]^2+D[3]^2-D[4]^2, D[2]^2, -D[2]^2, -2*D[1]*D[2],
> -2*D[2]*D[3], 0, 0, 2*D[2]*D[4], 0],
> [D[3]^2, D[3]^2, D[1]^2+D[2]^2-D[4]^2, -D[3]^2, 0, -2*D[2]*D[3],
> 2*D[3]*D[4], -2*D[1]*D[3], 0, 0],
> [D[4]^2, D[4]^2, D[4]^2, D[1]^2+D[2]^2+D[3]^2, 0, 0, -2*D[3]*D[4], 0,
> -2*D[2]*D[4], -2*D[1]*D[4]],
> [0, 0, D[1]*D[2], -D[1]*D[2], D[3]^2-D[4]^2, -D[1]*D[3], 0,
> -D[2]*D[3], D[1]*D[4], D[2]*D[4]],
> [D[2]*D[3], 0, 0, -D[2]*D[3],-D[1]*D[3], D[1]^2-D[4]^2, D[2]*D[4],
> -D[1]*D[2], D[3]*D[4], 0],
> [D[3]*D[4], D[3]*D[4], 0, 0, 0, -D[2]*D[4], D[1]^2+D[2]^2,
> -D[1]*D[4], -D[2]*D[3], -D[1]*D[3]],
> [0, D[1]*D[3], 0, -D[1]*D[3], -D[2]*D[3], -D[1]*D[2], D[1]*D[4],
> D[2]^2-D[4]^2, 0, D[3]*D[4]],
> [D[2]*D[4], 0, D[2]*D[4], 0, -D[1]*D[4], -D[3]*D[4], -D[2]*D[3], 0,
> D[1]^2+D[3]^2, -D[1]*D[2]],
> [0, D[1]*D[4], D[1]*D[4], 0, -D[2]*D[4], 0, -D[1]*D[3], -D[3]*D[4],
> -D[1]*D[2], D[2]^2+D[3]^2]]);
```

$$R := \left[\begin{array}{c}
D_2{}^2 + D_3{}^2 - D_4{}^2 ,\, D_1{}^2 ,\, D_1{}^2 ,\, -D_1{}^2 ,\, -2\,D_1\,D_2 ,\, 0 ,\, 0 ,\, -2\,D_1\,D_3 ,\, 0 ,\, 2\,D_1\,D_4 \\
D_2{}^2 ,\, D_1{}^2 + D_3{}^2 - D_4{}^2 ,\, D_2{}^2 ,\, -D_2{}^2 ,\, -2\,D_1\,D_2 ,\, -2\,D_2\,D_3 ,\, 0 ,\, 0 ,\, 2\,D_2\,D_4 ,\, 0 \\
D_3{}^2 ,\, D_3{}^2 ,\, D_1{}^2 + D_2{}^2 - D_4{}^2 ,\, -D_3{}^2 ,\, 0 ,\, -2\,D_2\,D_3 ,\, 2\,D_3\,D_4 ,\, -2\,D_1\,D_3 ,\, 0 ,\, 0 \\
D_4{}^2 ,\, D_4{}^2 ,\, D_4{}^2 ,\, D_1{}^2 + D_2{}^2 + D_3{}^2 ,\, 0 ,\, 0 ,\, -2\,D_3\,D_4 ,\, 0 ,\, -2\,D_2\,D_4 ,\, -2\,D_1\,D_4 \\
0 ,\, 0 ,\, D_1\,D_2 ,\, -D_1\,D_2 ,\, D_3{}^2 - D_4{}^2 ,\, -D_1\,D_3 ,\, 0 ,\, -D_2\,D_3 ,\, D_1\,D_4 ,\, D_2\,D_4 \\
D_2\,D_3 ,\, 0 ,\, 0 ,\, -D_2\,D_3 ,\, -D_1\,D_3 ,\, D_1{}^2 - D_4{}^2 ,\, D_2\,D_4 ,\, -D_1\,D_2 ,\, D_3\,D_4 ,\, 0 \\
D_3\,D_4 ,\, D_3\,D_4 ,\, 0 ,\, 0 ,\, 0 ,\, -D_2\,D_4 ,\, D_1{}^2 + D_2{}^2 ,\, -D_1\,D_4 ,\, -D_2\,D_3 ,\, -D_1\,D_3 \\
0 ,\, D_1\,D_3 ,\, 0 ,\, -D_1\,D_3 ,\, -D_2\,D_3 ,\, -D_1\,D_2 ,\, D_1\,D_4 ,\, D_2{}^2 - D_4{}^2 ,\, 0 ,\, D_3\,D_4 \\
D_2\,D_4 ,\, 0 ,\, D_2\,D_4 ,\, 0 ,\, -D_1\,D_4 ,\, -D_3\,D_4 ,\, -D_2\,D_3 ,\, 0 ,\, D_1{}^2 + D_3{}^2 ,\, -D_1\,D_2 \\
0 ,\, D_1\,D_4 ,\, D_1\,D_4 ,\, 0 ,\, -D_2\,D_4 ,\, 0 ,\, -D_1\,D_3 ,\, -D_3\,D_4 ,\, -D_1\,D_2 ,\, D_2{}^2 + D_3{}^2
\end{array}\right]$$

First of all, let us compute the rank of the *Alg*-module M associated with R.

```
> OreRank(R, Alg);
```

$$4$$

The formal adjoint of the linearized Ricci equations is defined by the following matrix of differential operators.

```
> R_adj := Involution(R, Alg);
```

$$
R\_adj := \left[\begin{array}{c}
{D_2}^2 + {D_3}^2 - {D_4}^2 , {D_2}^2 , {D_3}^2 , {D_4}^2 , 0 , D_2 D_3 , D_3 D_4 , 0 , D_2 D_4 , 0 \\
{D_1}^2 , {D_1}^2 + {D_3}^2 - {D_4}^2 , {D_3}^2 , {D_4}^2 , 0 , 0 , D_3 D_4 , D_1 D_3 , 0 , D_1 D_4 \\
{D_1}^2 , {D_2}^2 , {D_1}^2 + {D_2}^2 - {D_4}^2 , {D_4}^2 , D_1 D_2 , 0 , 0 , 0 , D_2 D_4 , D_1 D_4 \\
-{D_1}^2 , -{D_2}^2 , -{D_3}^2 , {D_1}^2 + {D_2}^2 + {D_3}^2 , -D_1 D_2 , -D_2 D_3 , 0 , -D_1 D_3 , 0 , 0 \\
-2 D_1 D_2 , -2 D_1 D_2 , 0 , 0 , {D_3}^2 - {D_4}^2 , -D_1 D_3 , 0 , -D_2 D_3 , -D_1 D_4 , -D_2 D_4 \\
0 , -2 D_2 D_3 , -2 D_2 D_3 , 0 , -D_1 D_3 , {D_1}^2 - {D_4}^2 , -D_2 D_4 , -D_1 D_2 , -D_3 D_4 , 0 \\
0 , 0 , 2 D_3 D_4 , -2 D_3 D_4 , 0 , D_2 D_4 , {D_1}^2 + {D_2}^2 , D_1 D_4 , -D_2 D_3 , -D_1 D_3 \\
-2 D_1 D_3 , 0 , -2 D_1 D_3 , 0 , -D_2 D_3 , -D_1 D_2 , -D_1 D_4 , {D_2}^2 - {D_4}^2 , 0 , -D_3 D_4 \\
0 , 2 D_2 D_4 , 0 , -2 D_2 D_4 , D_1 D_4 , D_3 D_4 , -D_2 D_3 , 0 , {D_1}^2 + {D_3}^2 , -D_1 D_2 \\
2 D_1 D_4 , 0 , 0 , -2 D_1 D_4 , D_2 D_4 , 0 , -D_1 D_3 , D_3 D_4 , -D_1 D_2 , {D_2}^2 + {D_3}^2
\end{array}\right]
$$

We compute the first extension module ext^1 with values in *Alg* of the formal adjoint of the linearized Ricci equations in *Alg*.

```
> st := time(): Ext1 := Exti(R_adj, Alg, 1): time() - st; Ext1[1];
```

$$44.639$$

$$
\left[\begin{array}{c}
\%1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \\
0, \%1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \\
0, 0, \%1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \\
0, 0, 0, \%1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \\
0, 0, 0, 0, \%1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \\
0, 0, 0, 0, 0, \%1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \\
0, 0, 0, 0, 0, 0, \%1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \\
0, 0, 0, 0, 0, 0, 0, \%1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \\
0, 0, 0, 0, 0, 0, 0, 0, \%1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \\
0, 0, 0, 0, 0, 0, 0, 0, 0, \%1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \\
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, \%1, 0, 0, 0, 0, 0, 0, 0, 0, 0 \\
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, \%1, 0, 0, 0, 0, 0, 0, 0, 0 \\
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, \%1, 0, 0, 0, 0, 0, 0, 0 \\
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, \%1, 0, 0, 0, 0, 0, 0 \\
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, \%1, 0, 0, 0, 0, 0 \\
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, \%1, 0, 0, 0, 0 \\
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, \%1, 0, 0, 0 \\
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, \%1, 0, 0 \\
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, \%1, 0 \\
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, \%1
\end{array}\right]
$$

$$\%1 := {D_1}^2 + {D_2}^2 - {D_4}^2 + {D_3}^2$$

```
> Ext1[2];
```

$$
\left[\begin{array}{c}
D_2 D_4 , 0 , 0 , 0 , -D_1 D_4 , 0 , 0 , 0 , {D_1}^2 , -D_1 D_2 \\
D_3 D_4 , 0 , 0 , 0 , 0 , 0 , {D_1}^2 , -D_1 D_4 , 0 , -D_1 D_3 \\
D_2 D_3 , 0 , 0 , 0 , -D_1 D_3 , {D_1}^2 , 0 , -D_1 D_2 , 0 , 0 \\
{D_4}^2 , 0 , 0 , {D_1}^2 , 0 , 0 , 0 , 0 , 0 , -2 D_1 D_4 \\
{D_3}^2 , 0 , {D_1}^2 , 0 , 0 , 0 , 0 , -2 D_1 D_3 , 0 , 0 \\
{D_2}^2 , {D_1}^2 , 0 , 0 , -2 D_1 D_2 , 0 , 0 , 0 , 0 , 0 \\
0 , D_3 D_4 , 0 , 0 , 0 , -D_2 D_4 , {D_2}^2 , 0 , -D_2 D_3 , 0 \\
0 , {D_4}^2 , 0 , {D_2}^2 , 0 , 0 , 0 , 0 , -2 D_2 D_4 , 0 \\
0 , {D_3}^2 , {D_2}^2 , 0 , 0 , -2 D_2 D_3 , 0 , 0 , 0 , 0 \\
0 , -D_1 D_4 , 0 , 0 , D_2 D_4 , 0 , 0 , 0 , D_1 D_2 , -{D_2}^2 \\
0 , -D_1 D_3 , 0 , 0 , D_2 D_3 , D_1 D_2 , 0 , -{D_2}^2 , 0 , 0 \\
0 , 0 , {D_4}^2 , {D_3}^2 , 0 , 0 , -2 D_3 D_4 , 0 , 0 , 0 \\
0 , 0 , -D_2 D_4 , 0 , 0 , D_3 D_4 , D_2 D_3 , 0 , -{D_3}^2 , 0 \\
0 , 0 , -D_1 D_4 , 0 , 0 , 0 , D_1 D_3 , D_3 D_4 , 0 , -{D_3}^2 \\
0 , 0 , D_1 D_2 , 0 , {D_3}^2 , -D_1 D_3 , 0 , -D_2 D_3 , 0 , 0 \\
0 , 0 , 0 , D_2 D_3 , 0 , {D_4}^2 , -D_2 D_4 , 0 , -D_3 D_4 , 0 \\
0 , 0 , 0 , D_1 D_3 , 0 , 0 , -D_1 D_4 , {D_4}^2 , 0 , -D_3 D_4 \\
0 , 0 , 0 , D_1 D_2 , {D_4}^2 , 0 , 0 , 0 , -D_1 D_4 , -D_2 D_4 \\
0 , 0 , 0 , 0 , D_3 D_4 , -D_1 D_4 , D_1 D_2 , 0 , 0 , -D_2 D_3 \\
0 , 0 , 0 , 0 , 0 , -D_1 D_4 , 0 , D_2 D_4 , D_1 D_3 , -D_2 D_3
\end{array}\right]
$$

Interpretation of the result:

The second matrix *Ext1*[2] gives a family of generators of 20 torsion elements in the *Alg*-module M associated with the linearized Ricci equations. The first matrix *Ext1*[1] shows that all the torsion elements are killed by the same differential operator, namely the Dalembertian ${D_1}^2 - {D_4}^2 + {D_2}^2 + {D_3}^2$. The third matrix gives a parametrization of the torsion-free module $M \ / \ t(M)$.

In conclusion, the linearized Ricci equations cannot be parametrized, i.e., they do not admit a potential.

Let us rewrite the torsion elements in terms of the system variable y_1, ..., y_{10}:

```
>  TorsionElements(R, [seq(y[i](x[1],x[2],x[3],x[4]), i=1..10)], Alg);
```

$$
\left[\left[\begin{array}{c}
(\frac{\partial^2}{\partial x_2^2} \%20) + (\frac{\partial^2}{\partial x_3^2} \%20) - (\frac{\partial^2}{\partial x_4^2} \%20) + (\frac{\partial^2}{\partial x_1^2} \%20) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%19) + (\frac{\partial^2}{\partial x_3^2} \%19) - (\frac{\partial^2}{\partial x_4^2} \%19) + (\frac{\partial^2}{\partial x_1^2} \%19) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%18) + (\frac{\partial^2}{\partial x_3^2} \%18) - (\frac{\partial^2}{\partial x_4^2} \%18) + (\frac{\partial^2}{\partial x_1^2} \%18) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%17) + (\frac{\partial^2}{\partial x_3^2} \%17) - (\frac{\partial^2}{\partial x_4^2} \%17) + (\frac{\partial^2}{\partial x_1^2} \%17) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%16) + (\frac{\partial^2}{\partial x_3^2} \%16) - (\frac{\partial^2}{\partial x_4^2} \%16) + (\frac{\partial^2}{\partial x_1^2} \%16) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%15) + (\frac{\partial^2}{\partial x_3^2} \%15) - (\frac{\partial^2}{\partial x_4^2} \%15) + (\frac{\partial^2}{\partial x_1^2} \%15) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%14) + (\frac{\partial^2}{\partial x_3^2} \%14) - (\frac{\partial^2}{\partial x_4^2} \%14) + (\frac{\partial^2}{\partial x_1^2} \%14) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%13) + (\frac{\partial^2}{\partial x_3^2} \%13) - (\frac{\partial^2}{\partial x_4^2} \%13) + (\frac{\partial^2}{\partial x_1^2} \%13) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%12) + (\frac{\partial^2}{\partial x_3^2} \%12) - (\frac{\partial^2}{\partial x_4^2} \%12) + (\frac{\partial^2}{\partial x_1^2} \%12) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%11) + (\frac{\partial^2}{\partial x_3^2} \%11) - (\frac{\partial^2}{\partial x_4^2} \%11) + (\frac{\partial^2}{\partial x_1^2} \%11) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%10) + (\frac{\partial^2}{\partial x_3^2} \%10) - (\frac{\partial^2}{\partial x_4^2} \%10) + (\frac{\partial^2}{\partial x_1^2} \%10) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%9) + (\frac{\partial^2}{\partial x_3^2} \%9) - (\frac{\partial^2}{\partial x_4^2} \%9) + (\frac{\partial^2}{\partial x_1^2} \%9) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%8) + (\frac{\partial^2}{\partial x_3^2} \%8) - (\frac{\partial^2}{\partial x_4^2} \%8) + (\frac{\partial^2}{\partial x_1^2} \%8) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%7) + (\frac{\partial^2}{\partial x_3^2} \%7) - (\frac{\partial^2}{\partial x_4^2} \%7) + (\frac{\partial^2}{\partial x_1^2} \%7) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%6) + (\frac{\partial^2}{\partial x_3^2} \%6) - (\frac{\partial^2}{\partial x_4^2} \%6) + (\frac{\partial^2}{\partial x_1^2} \%6) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%5) + (\frac{\partial^2}{\partial x_3^2} \%5) - (\frac{\partial^2}{\partial x_4^2} \%5) + (\frac{\partial^2}{\partial x_1^2} \%5) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%4) + (\frac{\partial^2}{\partial x_3^2} \%4) - (\frac{\partial^2}{\partial x_4^2} \%4) + (\frac{\partial^2}{\partial x_1^2} \%4) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%3) + (\frac{\partial^2}{\partial x_3^2} \%3) - (\frac{\partial^2}{\partial x_4^2} \%3) + (\frac{\partial^2}{\partial x_1^2} \%3) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%2) + (\frac{\partial^2}{\partial x_3^2} \%2) - (\frac{\partial^2}{\partial x_4^2} \%2) + (\frac{\partial^2}{\partial x_1^2} \%2) = 0 \\
(\frac{\partial^2}{\partial x_2^2} \%1) + (\frac{\partial^2}{\partial x_3^2} \%1) - (\frac{\partial^2}{\partial x_4^2} \%1) + (\frac{\partial^2}{\partial x_1^2} \%1) = 0
\end{array} \right], \right.
$$

$\%1 := \theta_{20}(x_1, x_2, x_3, x_4)$
$\%2 := \theta_{19}(x_1, x_2, x_3, x_4)$
$\%3 := \theta_{18}(x_1, x_2, x_3, x_4)$
$\%4 := \theta_{17}(x_1, x_2, x_3, x_4)$
$\%5 := \theta_{16}(x_1, x_2, x_3, x_4)$
$\%6 := \theta_{15}(x_1, x_2, x_3, x_4)$
$\%7 := \theta_{14}(x_1, x_2, x_3, x_4)$
$\%8 := \theta_{13}(x_1, x_2, x_3, x_4)$
$\%9 := \theta_{12}(x_1, x_2, x_3, x_4)$
$\%10 := \theta_{11}(x_1, x_2, x_3, x_4)$
$\%11 := \theta_{10}(x_1, x_2, x_3, x_4)$
$\%12 := \theta_{9}(x_1, x_2, x_3, x_4)$
$\%13 := \theta_{8}(x_1, x_2, x_3, x_4)$
$\%14 := \theta_{7}(x_1, x_2, x_3, x_4)$
$\%15 := \theta_{6}(x_1, x_2, x_3, x_4)$
$\%16 := \theta_{5}(x_1, x_2, x_3, x_4)$
$\%17 := \theta_{4}(x_1, x_2, x_3, x_4)$
$\%18 := \theta_{3}(x_1, x_2, x_3, x_4)$
$\%19 := \theta_{2}(x_1, x_2, x_3, x_4)$
$\%20 := \theta_{1}(x_1, x_2, x_3, x_4)$

$$
\left[\begin{array}{c}
\theta_1(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_4\,\partial x_2}\,\%10)-(\frac{\partial^2}{\partial x_4\,\partial x_1}\,\%6)+(\frac{\partial^2}{\partial x_1{}^2}\,\%2)-(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\%1)\\
\theta_2(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_4\,\partial x_3}\,\%10)+(\frac{\partial^2}{\partial x_1{}^2}\,\%5)-(\frac{\partial^2}{\partial x_4\,\partial x_1}\,\%3)-(\frac{\partial^2}{\partial x_3\,\partial x_1}\,\%1)\\
\theta_3(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_3\,\partial x_2}\,\%10)-(\frac{\partial^2}{\partial x_3\,\partial x_1}\,\%6)+(\frac{\partial^2}{\partial x_1{}^2}\,\%4)-(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\%3)\\
\theta_4(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_4{}^2}\,\%10)+(\frac{\partial^2}{\partial x_1{}^2}\,\%7)-2\,(\frac{\partial^2}{\partial x_4\,\partial x_1}\,\%1)\\
\theta_5(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_3{}^2}\,\%10)+(\frac{\partial^2}{\partial x_1{}^2}\,\%8)-2\,(\frac{\partial^2}{\partial x_3\,\partial x_1}\,\%3)\\
\theta_6(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_2{}^2}\,\%10)+(\frac{\partial^2}{\partial x_1{}^2}\,\%9)-2\,(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\%6)\\
\theta_7(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_4\,\partial x_3}\,\%9)-(\frac{\partial^2}{\partial x_4\,\partial x_2}\,\%4)+(\frac{\partial^2}{\partial x_2{}^2}\,\%5)-(\frac{\partial^2}{\partial x_3\,\partial x_2}\,\%2)\\
\theta_8(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_4{}^2}\,\%9)+(\frac{\partial^2}{\partial x_2{}^2}\,\%7)-2\,(\frac{\partial^2}{\partial x_4\,\partial x_2}\,\%2)\\
\theta_9(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_3{}^2}\,\%9)+(\frac{\partial^2}{\partial x_2{}^2}\,\%8)-2\,(\frac{\partial^2}{\partial x_3\,\partial x_2}\,\%4)\\
\theta_{10}(x_1,\,x_2,\,x_3,\,x_4)=-(\frac{\partial^2}{\partial x_4\,\partial x_1}\,\%9)+(\frac{\partial^2}{\partial x_4\,\partial x_2}\,\%6)+(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\%2)-(\frac{\partial^2}{\partial x_2{}^2}\,\%1)\\
\theta_{11}(x_1,\,x_2,\,x_3,\,x_4)=-(\frac{\partial^2}{\partial x_3\,\partial x_1}\,\%9)+(\frac{\partial^2}{\partial x_3\,\partial x_2}\,\%6)+(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\%4)-(\frac{\partial^2}{\partial x_2{}^2}\,\%3)\\
\theta_{12}(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_4{}^2}\,\%8)+(\frac{\partial^2}{\partial x_3{}^2}\,\%7)-2\,(\frac{\partial^2}{\partial x_4\,\partial x_3}\,\%5)\\
\theta_{13}(x_1,\,x_2,\,x_3,\,x_4)=-(\frac{\partial^2}{\partial x_4\,\partial x_2}\,\%8)+(\frac{\partial^2}{\partial x_4\,\partial x_3}\,\%4)+(\frac{\partial^2}{\partial x_3\,\partial x_2}\,\%5)-(\frac{\partial^2}{\partial x_3{}^2}\,\%2)\\
\theta_{14}(x_1,\,x_2,\,x_3,\,x_4)=-(\frac{\partial^2}{\partial x_4\,\partial x_1}\,\%8)+(\frac{\partial^2}{\partial x_3\,\partial x_1}\,\%5)+(\frac{\partial^2}{\partial x_4\,\partial x_3}\,\%3)-(\frac{\partial^2}{\partial x_3{}^2}\,\%1)\\
\theta_{15}(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\%8)+(\frac{\partial^2}{\partial x_3{}^2}\,\%6)-(\frac{\partial^2}{\partial x_3\,\partial x_1}\,\%4)-(\frac{\partial^2}{\partial x_3\,\partial x_2}\,\%3)\\
\theta_{16}(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_3\,\partial x_2}\,\%7)+(\frac{\partial^2}{\partial x_4{}^2}\,\%4)-(\frac{\partial^2}{\partial x_4\,\partial x_2}\,\%5)-(\frac{\partial^2}{\partial x_4\,\partial x_3}\,\%2)\\
\theta_{17}(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_3\,\partial x_1}\,\%7)-(\frac{\partial^2}{\partial x_4\,\partial x_1}\,\%5)+(\frac{\partial^2}{\partial x_4{}^2}\,\%3)-(\frac{\partial^2}{\partial x_4\,\partial x_3}\,\%1)\\
\theta_{18}(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\%7)+(\frac{\partial^2}{\partial x_4{}^2}\,\%6)-(\frac{\partial^2}{\partial x_4\,\partial x_1}\,\%2)-(\frac{\partial^2}{\partial x_4\,\partial x_2}\,\%1)\\
\theta_{19}(x_1,\,x_2,\,x_3,\,x_4)=(\frac{\partial^2}{\partial x_4\,\partial x_3}\,\%6)-(\frac{\partial^2}{\partial x_4\,\partial x_1}\,\%4)+(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\%5)-(\frac{\partial^2}{\partial x_3\,\partial x_2}\,\%1)\\
\theta_{20}(x_1,\,x_2,\,x_3,\,x_4)=-(\frac{\partial^2}{\partial x_4\,\partial x_1}\,\%4)+(\frac{\partial^2}{\partial x_4\,\partial x_2}\,\%3)+(\frac{\partial^2}{\partial x_3\,\partial x_1}\,\%2)-(\frac{\partial^2}{\partial x_3\,\partial x_2}\,\%1)
\end{array}\right]
$$

$$\%1 := y_{10}(x_1,\,x_2,\,x_3,\,x_4)$$
$$\%2 := y_9(x_1,\,x_2,\,x_3,\,x_4)$$
$$\%3 := y_8(x_1,\,x_2,\,x_3,\,x_4)$$
$$\%4 := y_6(x_1,\,x_2,\,x_3,\,x_4)$$
$$\%5 := y_7(x_1,\,x_2,\,x_3,\,x_4)$$
$$\%6 := y_5(x_1,\,x_2,\,x_3,\,x_4)$$
$$\%7 := y_4(x_1,\,x_2,\,x_3,\,x_4)$$
$$\%8 := y_3(x_1,\,x_2,\,x_3,\,x_4)$$
$$\%9 := y_2(x_1,\,x_2,\,x_3,\,x_4)$$
$$\%10 := y_1(x_1,\,x_2,\,x_3,\,x_4)$$

The parametrization of the torsion-free *Alg*-module $M\ /\ \mathrm{t}(M)$ is defined by:

```
> Ext1[3];
```

$$\begin{bmatrix} -2\,D_1 & 0 & 0 & 0 \\ 0 & -2\,D_2 & 0 & 0 \\ 0 & 0 & -2\,D_3 & 0 \\ 0 & 0 & 0 & -2\,D_4 \\ -D_2 & -D_1 & 0 & 0 \\ 0 & -D_3 & -D_2 & 0 \\ 0 & 0 & -D_4 & -D_3 \\ -D_3 & 0 & -D_1 & 0 \\ 0 & -D_4 & 0 & -D_2 \\ -D_4 & 0 & 0 & -D_1 \end{bmatrix}$$

Let us point out that this parametrization *Ext1* [3] depends on 4 arbitrary potentials. As the rank of the *Alg*-module M is also 4, and thus, the rank of the Alg-module $M\ /\ \mathrm{t}(M)$ is also 4, we deduce that *Ext1* [3] is a minimal parametrization of $M\ /\ \mathrm{t}(M)$.

We compute the second extension module extˆ2 with values in *Alg* of the formal adjoint of the linearized Ricci equations in *Alg*. The interpretation is the same as before.

```
> st := time():  Ext2 := Exti(R_adj, Alg, 2):  time() - st; Ext2[1];
```

$$2.839$$

$$\begin{bmatrix} D_1 & 0 & 0 & 0 \\ {D_4}^2 & 0 & 0 & 0 \\ D_3\,D_4 & 0 & 0 & 0 \\ D_2\,D_4 & 0 & 0 & 0 \\ {D_3}^2 & 0 & 0 & 0 \\ D_2\,D_3 & 0 & 0 & 0 \\ {D_2}^2 & 0 & 0 & 0 \\ 0 & D_2 & 0 & 0 \\ 0 & {D_4}^2 & 0 & 0 \\ 0 & D_3\,D_4 & 0 & 0 \\ 0 & D_1\,D_4 & 0 & 0 \\ 0 & {D_3}^2 & 0 & 0 \\ 0 & D_1\,D_3 & 0 & 0 \\ 0 & {D_1}^2 & 0 & 0 \\ 0 & 0 & D_3 & 0 \\ 0 & 0 & {D_4}^2 & 0 \\ 0 & 0 & D_2\,D_4 & 0 \\ 0 & 0 & D_1\,D_4 & 0 \\ 0 & 0 & {D_2}^2 & 0 \\ 0 & 0 & D_1\,D_2 & 0 \\ 0 & 0 & {D_1}^2 & 0 \\ 0 & 0 & 0 & D_4 \\ 0 & 0 & 0 & {D_3}^2 \\ 0 & 0 & 0 & D_2\,D_3 \\ 0 & 0 & 0 & D_1\,D_3 \\ 0 & 0 & 0 & {D_2}^2 \\ 0 & 0 & 0 & D_1\,D_2 \\ 0 & 0 & 0 & {D_1}^2 \end{bmatrix}$$

We compute a free resolution of the linearized Ricci equations.

```
> st:=time():  FreeResolution(R, Alg); time()-st;
```

$$
\mathrm{table}([1 = \left[\begin{array}{c}
D_2^2 + D_3^2 - D_4^2 ,\ D_1^2 ,\ D_1^2 ,\ -D_1^2 ,\ -2\,D_1\,D_2 ,\ 0 ,\ 0 ,\ -2\,D_1\,D_3 ,\ 0 ,\ 2\,D_1\,D_4 \\
D_2^2 ,\ D_1^2 + D_3^2 - D_4^2 ,\ D_2^2 ,\ -D_2^2 ,\ -2\,D_1\,D_2 ,\ -2\,D_2\,D_3 ,\ 0 ,\ 0 ,\ 2\,D_2\,D_4 ,\ 0 \\
D_3^2 ,\ D_3^2 ,\ D_1^2 + D_2^2 - D_4^2 ,\ -D_3^2 ,\ 0 ,\ -2\,D_2\,D_3 ,\ 2\,D_3\,D_4 ,\ -2\,D_1\,D_3 ,\ 0 ,\ 0 \\
D_4^2 ,\ D_4^2 ,\ D_4^2 ,\ D_1^2 + D_2^2 + D_3^2 ,\ 0 ,\ 0 ,\ -2\,D_3\,D_4 ,\ 0 ,\ -2\,D_2\,D_4 ,\ -2\,D_1\,D_4 \\
0 ,\ 0 ,\ D_1\,D_2 ,\ -D_1\,D_2 ,\ D_3^2 - D_4^2 ,\ -D_1\,D_3 ,\ 0 ,\ -D_2\,D_3 ,\ D_1\,D_4 ,\ D_2\,D_4 \\
D_2\,D_3 ,\ 0 ,\ 0 ,\ -D_2\,D_3 ,\ -D_1\,D_3 ,\ D_1^2 - D_4^2 ,\ D_2\,D_4 ,\ -D_1\,D_2 ,\ D_3\,D_4 ,\ 0 \\
D_3\,D_4 ,\ D_3\,D_4 ,\ 0 ,\ 0 ,\ 0 ,\ -D_2\,D_4 ,\ D_1^2 + D_2^2 ,\ -D_1\,D_4 ,\ -D_2\,D_3 ,\ -D_1\,D_3 \\
0 ,\ D_1\,D_3 ,\ 0 ,\ -D_1\,D_3 ,\ -D_2\,D_3 ,\ -D_1\,D_2 ,\ D_1\,D_4 ,\ D_2^2 - D_4^2 ,\ 0 ,\ D_3\,D_4 \\
D_2\,D_4 ,\ 0 ,\ D_2\,D_4 ,\ 0 ,\ -D_1\,D_4 ,\ -D_3\,D_4 ,\ -D_2\,D_3 ,\ 0 ,\ D_1^2 + D_3^2 ,\ -D_1\,D_2 \\
0 ,\ D_1\,D_4 ,\ D_1\,D_4 ,\ 0 ,\ -D_2\,D_4 ,\ 0 ,\ -D_1\,D_3 ,\ -D_3\,D_4 ,\ -D_1\,D_2 ,\ D_2^2 + D_3^2
\end{array}\right],
$$

$$
2 = \left[\begin{array}{cccccccccc}
-D_4 & -D_4 & -D_4 & -D_4 & 0 & 0 & 2\,D_3 & 0 & 2\,D_2 & 2\,D_1 \\
-D_3 & -D_3 & D_3 & D_3 & 0 & 2\,D_2 & -2\,D_4 & 2\,D_1 & 0 & 0 \\
-D_2 & D_2 & -D_2 & D_2 & 2\,D_1 & 2\,D_3 & 0 & 0 & -2\,D_4 & 0 \\
D_1 & -D_1 & -D_1 & D_1 & 2\,D_2 & 0 & 0 & 2\,D_3 & 0 & -2\,D_4
\end{array}\right],
$$

$3 = \mathrm{INJ}(4)$

])

4.190

We compute a free resolution of the formal adjoint of the linearized Ricci equations.

```
> st:=time(): FreeResolution(R_adj, Alg); time()-st;
```

$$
\mathrm{table}([1 = \left[\begin{array}{c}
D_2^2 + D_3^2 - D_4^2 ,\ D_2^2 ,\ D_3^2 ,\ D_4^2 ,\ 0 ,\ D_2\,D_3 ,\ D_3\,D_4 ,\ 0 ,\ D_2\,D_4 ,\ 0 \\
D_1^2 ,\ D_1^2 + D_3^2 - D_4^2 ,\ D_3^2 ,\ D_4^2 ,\ 0 ,\ 0 ,\ D_3\,D_4 ,\ D_1\,D_3 ,\ 0 ,\ D_1\,D_4 \\
D_1^2 ,\ D_2^2 ,\ D_1^2 + D_2^2 - D_4^2 ,\ D_4^2 ,\ D_1\,D_2 ,\ 0 ,\ 0 ,\ 0 ,\ D_2\,D_4 ,\ D_1\,D_4 \\
-D_1^2 ,\ -D_2^2 ,\ -D_3^2 ,\ D_1^2 + D_2^2 + D_3^2 ,\ -D_1\,D_2 ,\ -D_2\,D_3 ,\ 0 ,\ -D_1\,D_3 ,\ 0 ,\ 0 \\
-2\,D_1\,D_2 ,\ -2\,D_1\,D_2 ,\ 0 ,\ 0 ,\ D_3^2 - D_4^2 ,\ -D_1\,D_3 ,\ 0 ,\ -D_2\,D_3 ,\ -D_1\,D_4 ,\ -D_2\,D_4 \\
0 ,\ -2\,D_2\,D_3 ,\ -2\,D_2\,D_3 ,\ 0 ,\ -D_1\,D_3 ,\ D_1^2 - D_4^2 ,\ -D_2\,D_4 ,\ -D_1\,D_2 ,\ -D_3\,D_4 ,\ 0 \\
0 ,\ 0 ,\ 2\,D_3\,D_4 ,\ -2\,D_3\,D_4 ,\ 0 ,\ D_2\,D_4 ,\ D_1^2 + D_2^2 ,\ D_1\,D_4 ,\ -D_2\,D_3 ,\ -D_1\,D_3 \\
-2\,D_1\,D_3 ,\ 0 ,\ -2\,D_1\,D_3 ,\ 0 ,\ -D_2\,D_3 ,\ -D_1\,D_2 ,\ -D_1\,D_4 ,\ D_2^2 - D_4^2 ,\ 0 ,\ -D_3\,D_4 \\
0 ,\ 2\,D_2\,D_4 ,\ 0 ,\ -2\,D_2\,D_4 ,\ D_1\,D_4 ,\ D_3\,D_4 ,\ -D_2\,D_3 ,\ 0 ,\ D_1^2 + D_3^2 ,\ -D_1\,D_2 \\
2\,D_1\,D_4 ,\ 0 ,\ 0 ,\ -2\,D_1\,D_4 ,\ D_2\,D_4 ,\ 0 ,\ -D_1\,D_3 ,\ D_3\,D_4 ,\ -D_1\,D_2 ,\ D_2^2 + D_3^2
\end{array}\right],
$$

$$
2 = \left[\begin{array}{cccccccccc}
2\,D_1 & 0 & 0 & 0 & D_2 & 0 & 0 & D_3 & 0 & D_4 \\
0 & 2\,D_2 & 0 & 0 & D_1 & D_3 & 0 & 0 & D_4 & 0 \\
0 & 0 & 2\,D_3 & 0 & 0 & D_2 & D_4 & D_1 & 0 & 0 \\
0 & 0 & 0 & 2\,D_4 & 0 & 0 & D_3 & 0 & D_2 & D_1
\end{array}\right],
$$

$3 = \mathrm{INJ}(4)$

])

1.741